Solution du TP Maple n°1

1. Polynômes

```
> restart:
> A:=X^7-11*X^3+5; B:=X^3+7*X+1;
```

$$A := X^7 - 11\,X^3 + 5$$

$$B := X^3 + 7\,X + 1$$

```
> # première méthode :
> P:=add(a[i]*X^i,i=0..9):
> eqs:=coeffs(rem(P+1,A,X),X),coeffs(rem(P+2,B,X),X):
> vars:=coeffs(P,X):
> sols:=solve({eqs},{vars}):
> Psol:=subs(sols,P);
```

$$Psol := -\frac{5080081}{2554951} + \frac{44705}{2554951}\,X - \frac{360295}{2554951}\,X^2 + \frac{5555286}{2554951}\,X^3 - \frac{98351}{2554951}\,X^4 + \frac{792649}{2554951}\,X^5 - \frac{505026}{2554951}\,X^7 + \frac{8941}{2554951}\,X^8 - \frac{72059}{2554951}\,X^9$$

```
> # seconde méthode :
> gcdex(A,B,X,'U','V');
```

$$1$$

```
> Psolbis:=expand(-A*U-1);
```

$$Psolbis := -\frac{5080081}{2554951} + \frac{44705}{2554951}\,X - \frac{360295}{2554951}\,X^2 + \frac{5555286}{2554951}\,X^3 - \frac{98351}{2554951}\,X^4 + \frac{792649}{2554951}\,X^5 - \frac{505026}{2554951}\,X^7 + \frac{8941}{2554951}\,X^8 - \frac{72059}{2554951}\,X^9$$

```
> # vérification :
> Psol-Psolbis;
```

$$0$$

2. Polynômes (bis)

```
> restart:
> P:=X^5-3*X^3+2*X^2-1;
```

$$P := X^5 - 3\,X^3 + 2\,X^2 - 1$$

```
> rac:=solve(P):
> allvalues(rac[1]);
```

$$\mathrm{RootOf}\left(\_Z^5 - 3\,\_Z^3 + 2\,\_Z^2 - 1, index = 1\right)$$

```
> # Les racines ne se calculent pas.
> P1:=product(product(X-r-s,s=RootOf(P,X)),r=RootOf(P,X)):
> P2:=product(X-2*r,r=RootOf(P,X)):
> P3:=normal(P1/P2):
> Q:=psqrt(P3);
```

$$Q := 5 + X - 2\,X^2 + 6\,X^3 - 31\,X^4 - X^5 + 27\,X^6 + 2\,X^7 - 9\,X^8 + X^{10}$$

```
> # Expression générale de Q :
> P:=X^5+a*X^4+b*X^3+c*X^2+d*X+e;
```

$$P := X^5 + a X^4 + b X^3 + c X^2 + d X + e$$

```
> P1:=product(product(X-r-s,s=RootOf(P,X)),r=RootOf(P,X)):
> P2:=product(X-2*r,r=RootOf(P,X)):
> P3:=normal(P1/P2):
> Q:=collect(psqrt(P3),X);
```

$$Q := X^{10} + 4 a X^9 + (6 a^2 + 3 b) X^8 + (9 b a + c + 4 a^3) X^7 + (3 b^2 + 9 b a^2 + 4 c a - 3 d + a^4) X^6 + (-5 d a + 6 b^2 a + 2 b c + 5 a^2 c - 11 e + 3 b a^3) X^5 + (-c^2 + 3 b^2 a^2 - 2 a^2 d + 2 a^3 c + 6 b a c - 2 b d - 22 e a + b^3) X^4 + (-4 d c + b^3 a - 4 e b - 16 a^2 e + b^2 c + 4 b a^2 c) X^3 + (b a^2 d - 4 d^2 - b c^2 + 2 b^2 a c - 3 a d c - 9 b a e - 4 a^3 e + a^2 c^2 + b^2 d + 7 e c) X^2 + (b^2 a d + b c^2 a - 4 d^2 a + 4 d e - 4 b a^2 e + 4 a e c - b^2 e - c^3) X + b a d c + b e c - b^2 a e + 2 d a e - d c^2 - e^2 - a^2 d^2$$

3. Formule d'approximation de la dérivée

```
> restart:
> expr1:=(f(x0+k*h)-f(x0-k*h))/h;
```

$$expr1 := \frac{f(x0 + k h) - f(x0 - k h)}{h}$$

```
> n:=5:
> expr2:=add(a[k]*expr1,k=1..n);
```

$$expr2 := \frac{a_1 (f(x0 + h) - f(x0 - h))}{h} + \frac{a_2 (f(x0 + 2 h) - f(x0 - 2 h))}{h} + \frac{a_3 (f(x0 + 3 h) - f(x0 - 3 h))}{h} + \frac{a_4 (f(x0 + 4 h) - f(x0 - 4 h))}{h} + \frac{a_5 (f(x0 + 5 h) - f(x0 - 5 h))}{h}$$

```
> t:=taylor(expr2-D(f)(x0),h=0,2*n):
> t2:=convert(t,polynom):
> eqs:=seq(coeff(t2,h,i),i=0..2*n-1):
> s:=solve({eqs},{seq(a[i],i=1..n)}):
> approx:=subs(s,expr2);
```

$$approx := \frac{5}{6} \frac{f(x0 + h) - f(x0 - h)}{h} - \frac{5}{21} \frac{f(x0 + 2 h) - f(x0 - 2 h)}{h} + \frac{5}{84} \frac{f(x0 + 3 h) - f(x0 - 3 h)}{h} - \frac{5}{504} \frac{f(x0 + 4 h) - f(x0 - 4 h)}{h} + \frac{1}{1260} \frac{f(x0 + 5 h) - f(x0 - 5 h)}{h}$$

```
> # application numérique :
> f:=sin; x0:=1;
```

$$f := \sin$$

$$x0 := 1$$

```
> for h in [1,0.1,0.01,0.001] do evalf(D(f)(x0)-approx,50)
  end do;
```

$$0.00013686498758241081179279376716088591163607573045932$$

$$1.9422824387242667770330730114904619 33\ 10^{-14}$$

$$1.9490737989847134194169532 0\ 10^{-24}$$

$$1.9491418303642793\ 10^{-34}$$

4. Développement asymptotique

```
> restart:
> eq:=tan(x)=x;
```

$$eq := \tan(x) = x$$

```
> # un peu de gymnastique :
> eq1:=subs(x=n*Pi+u,eq);  # avec u entre 0 et Pi/2
```

$$eq1 := \tan(n\pi + u) = n\pi + u$$

```
> eq2:=map(arctan,eq1):
> eq3:=subs(tan(n*Pi+u)=tan(u),eq2):
> eq4:=simplify(eq3,symbolic);
```

$$eq4 := u = \arctan(n\pi + u)$$

```
> limit(subs(eq4,u),n=infinity);
```

$$\frac{1}{2}\pi$$

```
> # passons au développement asymptotique :
> xn:=n*Pi+Pi/2+a/n+b/n^2+c/n^3+d/n^4+e/n^5;
```

$$xn := n\pi + \frac{1}{2}\pi + \frac{a}{n} + \frac{b}{n^2} + \frac{c}{n^3} + \frac{d}{n^4} + \frac{e}{n^5}$$

```
> s1:=series(tan(xn-n*Pi)-xn,n=infinity,4):
> s2:=convert(s1,polynom):
> s3:=map(simplify,s2);
```

$$s3 := -\frac{(1+\pi a)\,n}{a} + \frac{b}{a^2} - \frac{1}{2}\pi - \frac{1}{3}\,\frac{2a^4 - 3ca + 3b^2}{a^3 n}$$
$$- \frac{1}{3}\,\frac{2ba^4 - 3da^2 + 6bca - 3b^3}{a^4 n^2}$$
$$+ \frac{1}{45}\,\frac{45ea^3 - 90bda^2 - 45c^2a^2 + 135cab^2 - 45b^4 - 30ca^5 + a^8}{a^5 n^3}$$

```
> eqs:=coeffs(s3,n):
> vars:=a,b,c,d,e:
> sols:=solve({eqs},{vars}):
> subs(sols,xn);
```

$$n\pi + \frac{1}{2}\pi - \frac{1}{\pi n} + \frac{1}{2\pi n^2} - \frac{1}{12}\,\frac{8+3\pi^2}{\pi^3 n^3} + \frac{1}{8}\,\frac{8+\pi^2}{\pi^3 n^4} - \frac{1}{240}\,\frac{240\pi^2 + 15\pi^4 + 208}{\pi^5 n^5}$$